

LIM

Version 0.5

User's Guide

January, 2024

Synclesis, Inc.

Urbana, IL

Copyright 2024 - Synclesis, Inc.

This version is the first major release of LIM. The development of LIM and its algorithms is an ongoing effort at Syncsis, and therefore not all of the intended capabilities of the simulator have been implemented in full yet.

1. INTRODUCTION

LIM is a general-purpose circuit simulation program. The current version focuses on nonlinear transient analysis. It will also predict DC behavior. Circuits may contain resistors, capacitors, inductors, mutual inductors, independent voltage and current sources. Four types of dependent sources are also supported. Systems with frequency-dependent behavior can be handled with controllable accuracy. This version supports lossy and lossless single or coupled transmission lines and multi-port macromodels described by their frequency-dependent network parameters.

LIM currently supports three of most common semiconductor devices: diodes, BJTs, and MOSFETs. The development of models for these devices is a work in progress. Current models supported are the Ebers-Moll model for BJTs and level 1 for MOSFETs.

1.1. TYPES OF ANALYSIS

1.1.1. Transient Analysis

The transient analysis in LIM is performed using the latency insertion algorithm in which voltage and current variables are computed in a leapfrog manner as a function of time. The same general algorithm is applied to all elements in a circuit

The time interval is specified on a .TRAN control line.

1.3. STABILITY

Although the algorithm used in LIM has been found to be reliable, in some cases it is not stable and leads to a diverging solution. When this failure occurs, erroneous values are produced.

Failure to achieve a stable solution can be caused by incorrect circuit connections or a time step that is too large. LIM uses a fictitious capacitor to ground for each node and a fictitious series inductor for some branches. Values for these fictitious elements can affect the stability, accuracy and speed of the simulation. The .LIM control line is used to control these parameters.

1.4. RUNNING LIM

The LIM simulation engine is invoked by issuing the command:

```
lim netlist directory index
```

Where netlist is the file name of the netlist containing the circuit description. The file name should include the full path from the current directory to the file of interest. directory is the name of the directory in which the output files resulting from the run will be stored.

Within the directory several files are created:

“*out_file*” which is a run summary indicating all the key properties of the network and of the simulation.

Symbol	Description
Nnodes	Total number of nodes
dtUser	Time step
dtMax	Maximum allowable time step
Npts	Number of points
Lsmal	Fictitious inductor
Csmal	Fictitious capacitor

“*data_file*” which tabulates the results of the transient simulation and list the time in one column and the voltages in subsequent column in the order in which they appear in the .print statement in the netlist.

“*lim_log*” a log file that contains some of the key parameters from the simulation.

“*netfile*” which contains the netlist with all the nodes renumbered.

index is a digit that can be 1 or 0

If index=0, only the set up is performed without simulation. Program halts before simulation. The summary of the setup is written in the file *out_file*.

If index=1, the complete simulation is performed.

2. NETLIST DESCRIPTION

2.1. General Structure and Conventions

LIM follows a nomenclature that is closely related to that of SPICE, version 3. Most of the format, conventions and netlist representations are the same as SPICE3.

The circuit to be analyzed is represented to LIM by a set of element lines describing the circuit and element values, and a set of control lines, that define the model parameters and the simulation parameters.

Lines starting with a “*” are comment lines and are ignored. The order in which the lines are listed is arbitrary and the last line in the netlist must be ".END". Continuation lines start with a “+”.

Each element in the circuit must be described by a line. Each element line contains the element name, the circuit nodes to which the element is connected, and the values of the parameters that determine the electrical characteristics of the element. The LIM element line starts with an alphanumeric string representing the element name and in which the first letter specifies the element type is denoted by with an alphanumeric string with the first letter of the element name specifying the element type. For example, a capacitor name must begin with the letter C and can contain one or more characters. As an example, C, C1, CIN, CLOAD, and C192A are valid capacitor names. Details of each type of device are supplied in a following section.

Items on a line are separated by one or more blanks. A comma, an equal (=) sign, or a left or right parenthesis; extra spaces are ignored. A line may be continued by entering a '+' (plus) in column 1 of the following line; LIM continues reading beginning with column 2.

A name item must begin with a letter (A through Z) and cannot contain any delimiters. A number may be an integer, a floating point. Exponent format (1e-14, 2.65e3) is also allowed.

Node names may be arbitrary character strings. The reference (ground) node must be named '0'.

2.2 DC Analysis

The dc solution in LIM is taken as the steady-state response resulting from all independent DC sources switching as steps (or ramps) to their respective values. It is in essence a solution to a transient simulation in which all variables eventually converge to constant values. In special cases involving oscillators, a constant DC value will not be reached; the DC analysis simulation will instead produce oscillations at a specific frequency.

2.3 Control Lines

2.3.1 .OP Line

The .OP line is used to control the parameters of the DC analysis. It indicates that only the DC analysis should be performed for the circuit. Execution will stop once the DC solution for the

network has been reached.

General form:

```
.OP TSTEP TSTOP
```

The DC operating point analysis in LIM is performed through a transient analysis. TSTEP is the time step used for that transient analysis. TSTOP is a time limit at which to stop the transient simulation. It is anticipated that at TSTOP, all the voltage and current values will converge to their final value which represent the DC operating point. For this analysis, all values start at zero. The DC independent sources are next ramped up from zero until they reached their DC values. After enough time has elapsed.

Examples:

```
.OP 1E-06 1E-09
```

2.3.2 .TRAN Line

The .TRAN line is used to control the parameters of the transient simulation

General form:

```
.TRAN TSTEP TSTOP
```

TSTEP is the time step used for the simulations. In this version of LIM, the time step is constant and must be small enough as to guarantee stability in the simulations. TSTOP is the duration of the simulation. The ratio TSTOP/TSTEP determines the number of transient simulation points.

Examples:

```
.TRAN 1E-06 1E-09
```

2.3.3 .PRINT Line

The .PRINT line is used to print the output of the simulation

General form:

```
.PRINT PRTYPE V(N1) V(N2) V(N3) V(N4)  
.PRINT PRTYPE I(EL1) I(EL2) I(EL3) I(EL4)
```

The .PRINT line defines the contents of a tabular listing of one to four output variables. PRTYPE is the type of analysis. For this version, only TRAN is allowed. The variables to be printed can be voltage or current.

- In the case of voltage, the node string value is specified between the parentheses.
- In the case of current, the name of the element appears between the parentheses. Some restrictions apply. The element must be a two-terminal element. Currently, resistors, capacitors, inductors, diodes, voltage and current sources (dependent or independent) are

allowed. The direction of the current is taken from the first to the second of the two-terminal element as specified on the netlist.

Examples:

```
.PRINT TRAN V(4) V(3)
.PRINT TRAN I(R4) I(L3)
```

2.3.4 .PLOT Lines

The .PRINT line is used to print the output of the simulation

General form:

```
.PLOT PRTYPE V(N1) V(N2) V(N3) V(N4)
.PLOT PRTYPE I(EL1) I(EL2) I(EL3) I(EL4)
```

The .PLOT line defines the contents of a tabular listing of one to four output variables. PRTYPE is the type of analysis. For this version, only TRAN is allowed. The variables to be plotted can be voltage or current.

- In the case of voltage, the node string value is specified between the parentheses.
- In the case of current, the name of the element appears between the parentheses. Some restrictions apply. The element must be a two-terminal element. Currently, resistors, capacitors, inductors, diodes, voltage and current sources (dependent or independent) are allowed. The direction of the current is taken from the first to the second of the two-terminal element as specified on the netlist.

Examples:

```
.PLOT TRAN V(4) V(3)
.PLOT TRAN I(R4) I(L3)
```

2.3.5 .LIM Line

Several parameters of the simulations available in LIM can be altered to control the accuracy, speed and stability. These parameters may be changed via the .LIM command.

General form

```
.LIM R=value G=value L= value C=value...
```

Each parameter refers to a LIM variable that controls one or several simulation constraints. They are defined below

C: Value of fictitious capacitance placed at each node of the network. Nodes that already have a capacitor to ground will ignore the fictitious capacitance. In general, it is best to choose a

value for C that is at least one order of magnitude smaller than the smallest capacitance in the network. As C decreases, accuracy increases for a given time step at the cost of stability.

L: Value of fictitious inductance placed at each branch of the network. Branches that are inductors will ignore the fictitious inductance. In general, it is best to choose a value for L that is at least one order of magnitude smaller than the smallest inductance in the network. As L decreases, accuracy increases for a given time step at the cost of stability.

G: Value of fictitious conductance placed at each node of the network. Nodes that already have a resistor to ground will ignore that fictitious conductance. In general, G is chosen to be very small. In some rare cases, a relatively larger value of G it may help facilitate the simulations at the cost of accuracy. This option is added to provide more flexibility in the simulations.

P: Frequency at which simulation variables that are to be printed will be printed.

O: Frequency at which simulation variables that are to be plotted will be plotted.

D: DC analysis control parameter. A value of 0 means that the DC analysis will be performed before the transient analysis. A value of 1 indicates that only the DC analysis will be performed.

S: Slew rate control parameter for DC analysis. The value assigned to S will be multiplied by the time step and used as the rise time of the step voltages and step currents for the DC analysis.

Example

```
.LIM C=1.010e-14 L=0.1e-09 G=1.0e-20 P=80 O=10 D=0
```

2.3.6 .MODEL Line - Device Models

The .MODEL line contains parameter values for special devices. For these device types, each device element line contains the device name, the nodes to which the device is connected, and the device model name.

General form:

```
.MODEL MNAME TYPE( PNAME1=PVAL1 PNAME2=PVAL2 . . . )
```

Example:

```
.MODEL MOD1 NPN (BF=50 IS=1E-13 VBF=50)
```

The .MODEL line specifies the parameter values for that particular element line. Several different elements may refer to the same model line with the same parameters. In that case, they must use the same model name (MNAME).

MNAME in the above is the model name. Model types currently available in this version are:

- LTRA: Lossy transmission line model

- D: Diode model
- NPN: NPN bipolar junction transistor model
- PNP: PNP bipolar junction transistor model
- NMOS: N-channel MOSFET model
- PMOS: P-channel MOSFET model

Parameter values are defined by appending the parameter name followed by an equal sign and the parameter value. Model parameters that are not given a value are assigned default values that depend on the model type. Model parameters and default values are listed in the respective sections along with the description of device element lines.

2.3.7 .SUBCKT Line - Subcircuits

A subcircuit is defined as a grouping of elements. A set of nodes is next defined to provide access to the subcircuit from external elements or other subcircuits. The group of elements within a subcircuit can be accessed wherever the subcircuit is referenced. There is no limit on the size or complexity of subcircuits, and subcircuits may contain other subcircuits.

General form:

```
.SUBCKT subcktname N1 N2 N3 ...
```

A circuit definition starts with a .SUBCKT line. The name of the subcircuit is SUBCKTNAME. N1, N2, ..are the external nodes; they cannot be zero. The element lines that follow represent the subcircuit. The last line is .ENDS . Control lines cannot be used within a subcircuit definition. Device model definitions can be invoked in a subcircuit block. In that case, they are strictly local and cannot be known outside of the subcircuit block. Except for the external nodes specified in the .SUBCKT line, all nodes are internal to the subcircuit block except for 0 (ground reference node) which is always a global node.

Example:

```
.SUBCKT OPAMP 1 2 3 4
```

2.3.7 Subcircuit Calls – X-element

To invoke a subcircuit in a LIM netlist, the X element is used. This consist of an element with an alphanumeric name that starts with X followed by the circuit nodes and the subcircuit name. The circuit nodes will map according to the order in which they were defined in the .SUBCKT definition for the subcircuit SUBCKTNAME.

General form:

```
XYZA N1 N2 N3 ... SUBCKTNAME
```

Example:

```
X1 2 4 17 3 1 MULTI
```


2.3.8 .MACR Line

The .MACR line is a special control line used to perform model-order reduction analysis on multi-port S-parameter data files. The .MACR command will override all other commands in the netlist and cause execution to terminate after the model-order reduction analysis.

General form:

```
.MACR FLAG
```

LIM will carry out the pole-residue extraction for all the N-elements of the netlist and exit right after without proceeding to the simulation steps. If FLAG is 0 passivity is not enforced in finding the poles and residues. If FLAG is 1, passivity is enforced.

Examples:

```
.MACR 1
```

3. CIRCUIT ELEMENTS AND MODELS

LIM is a circuit simulator that establishes the relationship between voltage and current variables through circuit elements. LIM distinguishes two types of elements. Simple elements and devices. Simple elements are two terminal constructs that define a branch. By convention, positive current flow is in the direction of voltage drop. Following is a sequence of simple elements.

1. Resistors - R- Element



General form:

```
RXYZ N1 N2 VALUE
```

Examples:

```
R1 1 2 100
```

```
RC1 12 17 1E+03
```

N1 and N2 are the two element nodes. VALUE is the resistance (in ohms).

2. Capacitors – C- Element



General form:

```
CXYZ N+ N- VALUE
```

Examples:

```
CBYP 13 0 1UF
COSC 17 23 10U IC=3V
```

N+ and N- are the positive and negative element nodes, respectively. VALUE is the capacitance in Farads.

3. Inductors – L-Element



General form:


```
LYZX N+ N- VALUE
```

Examples:

```
LLINK 42 69 1UH
LSHUNT 23 51 10U IC=15.7MA
```

N+ and N- are the positive and negative element nodes, respectively. VALUE is the inductance in Henries.

4. Coupled (Mutual) Inductors – K-Element



General form:

```
KXYZ LYZX LABC VALUE
```

Examples:

```
K43 LAA LBB 0.999
KXFRMR L1 L2 0.87
```

LYZX and LABC are the names of the two coupled inductors, and VALUE is the coefficient of coupling, K, which must be greater than 0 and less than or equal to 1. Using the 'dot' convention, place a 'dot' on the first node of each inductor.

5. Independent voltage sources – V-Element



DC Voltage Source:

General Form:

VXYZ N+ N- DC VALUE

N+ and N- are the positive and negative nodes, respectively. Note that voltage sources need not be grounded. DC indicates that it is a DC source. Value is the voltage in volts.

Example:

VCC 10 0 DC 6

Pulse Voltage Source

General Form:

VXYZ N+ N- PULSE(V1 V2 TD TR TF PW PER)

N+ and N- are the positive and negative nodes, respectively. PULSE indicates that it is a pulsed source. The pulse parameters are.

Parameter	Unit
V1 (initial value)	Volts
V2 (pulsed value)	Volts
TD (delay time)	seconds
TR (rise time)	seconds
TF (fall time)	seconds
PW (pulse width)	Seconds
PER (period)	seconds

Example:

VIN 3 0 PULSE(-1 1 2NS 2NS 2NS 50NS 100NS)

Sinusoidal Voltage Source

General Form:

VXYZ N+ N- SIN(VO VA FREQ TD THETA)

The signal has a sinusoidal shape with time. SIN indicates that it is a sinusoidal source. The sine parameters are

Parameter	Unit
VO (offset)	Volts
VA (amplitude)	Volts
FREQ (frequency)	Hz



TD (delay) 0.0 seconds
THETA (damping factor) 1/second

Example:

```
VIN 3 0 SIN(0 1 100MEG 1NS 1E10)
```

Exponential Voltage Source



General Form:

```
VXYZ N+ N- EXP(V1 V2 TD1 TAU1 TD2 TAU2)
```

EXP indicates that it is an exponential source. The exponential algorithm follows the function:

$$V_g = \begin{cases} V_1, & \text{for } 0 < t < t_{d1} \\ V_1 + (V_2 - V_1) \left[1 - e^{-\frac{(t-t_{d1})}{\tau_1}} \right], & \text{for } t_{d1} < t < t_{d2} \\ V_1 + (V_2 - V_1) \left[1 - e^{-\frac{(t-t_{d1})}{\tau_1}} \right] + (V_1 - V_2) \left[1 - e^{-\frac{(t-t_{d2})}{\tau_2}} \right], & \text{for } t > t_{d2} \end{cases}$$

The parameters are listed below

Parameter	Unit
V1 (initial value, V_1)	Volts
V2 (pulsed value, V_2)	Volts
TD1 (rise delay time, t_{d1})	seconds
TAU1 (rise time constant, τ_1)	seconds
TD2 (fall delay time, t_{d2})	second
TAU2 (fall time constant, τ_2)	seconds

Example:

```
VIN 3 0 EXP(-4 -1 2.0E-9 30.0E-9 60.0E-9 40.0E-9)
```

Gaussian Voltage Source

General Form:

VXYZ N+ N- GAUS(V1 TD TAU)

GAUS indicates that it is a Gaussian source. The Gaussian algorithm follows the function:

$$V_g = V_1 e^{-\frac{(t-t_d)^2}{2\tau^2}}$$

The parameters are listed below

Parameter	Unit
V1 (magnitudevalue, V_1)	Volts
TD (time delay time, t_d)	seconds
TAU (time constant, τ)	seconds

Example:

VIN 3 0 GAUS(1.0 2.0E-09 8.0E-09)

Surge Voltage Source

General Form:

VXYZ N+ N- SURG(V0 K TD A B)

SURG indicates that it is a voltage surge. The surge algorithm follows the equation:

$$V_g = \begin{cases} 0, & \text{for } t \leq t_d \\ V_o k \left(e^{-a(t-t_d)} - e^{-b(t-t_d)} \right), & \text{for } t \geq t_d \end{cases}$$

The parameters are listed below

Parameter	Unit
V0 (initial value, E_o)	Volts
K (attenuation factor, k)	-
TD (time delay)	seconds
A (rise time constant, a)	seconds
B (fall time constant, b)	second

Examples:

VIN 3 0 SURG(70 1.3 1.0e-09 4.0e+07 6.0e+08)



Piece-Wise Linear Voltage Source

General Form:

VXYZ N+ N- PWL(T1 V1 T2 V2 ... TK VK ... TN VN)

At each time T_i , the voltage at the source is V_i . Intermediate time and voltage points are obtained via interpolation.

Example:

VCK 7 5 PWL(0 -7 10.0e-9 -7 11.0e-9 -3 17.0e-9 -3
18.0e-9 -7 50.0e-9 -7)

Bit Sequence Voltage Source

General Form:

VXYZ N+ N- BITSEQ(V1 V2 XXXYYY BR TR TF)

This form indicates a bit sequence of zeros and ones. The maximum length is 2000. The sequence repeats itself. The parameters are listed below

Parameter	Unit
V1 (initial value,)	Volts
V2 (Final value)	Volts
XXXYYY	Bit Sequence
BR (bit rate)	Bits/seconds
TR (rise time)	seconds
TF (fall time)	second

Example:

VP 1 0 BITSEQ (0.33 1.8 0110011 1n 0.1n 1n)

User Defined Voltage Source

General Form:

VXYZ N+ N- USERDEF Filename

This form indicates a piecewise linear description of the voltage source can be read from a file. The points t_i and $v(t_i)$ are read directly from the file *Filename* in the format:

t_1 $v(t_1)$
 t_2 $v(t_2)$
 t_3 $v(t_3)$

The units are seconds and volts. During a run, if the time specified is larger than the last time point, then the voltage value corresponding to the last time point is returned.

Example:

```
VP 1 0 USERDEF voltage_listing.txt
```

Pseudo-Random Bit Sequence Voltage Source

General Form:

```
VXYZ N+ N- PRBS(V1 V2 BR TR TF)
```

This form indicates a bit sequence of zeros and ones that is randomly generated. The parameters are listed below

Parameter	Unit
V1 (initial value,)	volts
V2 (Final value)	volts
BR (bit rate)	bits/seconds
TR (rise time)	seconds
TF (fall time)	seconds

Example:

```
VP 1 0 PRBS (0.0 70 5.0e+08 1.0e-09 1.0e-09 )
```

6. Independent current sources – I-Element



DC Current Source:

General Form:

```
IXYZ N+ N- DC VALUE
```

N+ and N- are the positive and negative nodes, respectively. Positive current is assumed to flow from the positive node, through the source, to the negative node. A current source of positive value forces current to flow out of the N+ node, through the source, and into the N- node. DC indicates that it is a DC source. Value is the current in amperes.

Examples:

```
ICC 10 0 DC 6
```

Pulsed Current Source

General Form:

```
IXYZ N+ N- PULSE(I1 I2 TD TR TF PW PER)
```

Same convention applies for the parameters as in the pulse voltage source

case with voltage in volts replaced by current in amperes.

Examples

```
IIN 3 0 PULSE(-1 1 2NS 2NS 2NS 50NS 100NS)
```

Sinusoidal Current Source

General Form:

```
IXYZ N+ N- SIN(IO IA FREQ TD THETA)
```

Same convention applies for the parameters as in the sinusoidal voltage source case with voltage in volts replaced by current in amperes.

Examples:

```
IIN 3 0 SIN(0 1 100MEG 1NS 1E10)
```

Exponential Current Source

General Form:

```
IXYZ N+ N- EXP(I1 I2 TD1 TAU1 TD2 TAU2)
```

Same convention applies for the parameters as in the exponential voltage source case with voltage in volts replaced by current in amperes.

Examples:

```
IIN 3 0 EXP(-4 -1 2.0E-9 30.0E-9 60.0E-9 40.0E-9)
```

Gaussian Current Source

General Form:

```
IXYZ N+ N- GAUS(I1 TD TAU)
```

Same convention applies for the parameters as in the Gaussian voltage source case with voltage in volts replaced by current in amperes.

Example:

```
IIN 3 0 GAUS(1.0 2.0E-09 8.0E-09)
```

Surge Current Source

General Form:

```
IXYZ N+ N- SURG (IO K TD A B)
```

Same convention applies for the parameters as in the exponential voltage source case with voltage in volts replaced by current in amperes.

Examples:

```
IIN 3 0 SURG (70 1.3 1.0e-09 4.0e+07 6.0e+08)
```

Piecewise Linear Current Source

General Form:

```
IXYZ N+ N- PWL(T1 I1 T2 I2 ... TK IK ... TN IN)
```

Same convention applies for the parameters as in the piecewise linear voltage source case with voltage in volts replaced by current in amperes.

Example:

```
ICK 7 5 PWL(0 -7 10.0e-9 -7 11.0e-9 -3 17.0e-9 -3  
18.0e-9 -7 50.0e-9 -7)
```

Bit Sequence Current Source

General Form:

IXYZ N+ N- BITSEQ(I1 I2 XXXYYY BR TR TF)

This form indicates a bit sequence of zeros and ones. The sequence repeats itself. The parameters are listed below

Parameter	Unit
I1 (initial value,)	Amps
I2 (Final value)	Amps
XXXXYYY	Bit Sequence
BR (bit rate)	Bits/seconds
TR (rise time)	seconds
TF (fall time)	second

Example:

IP 1 0 BITSEQ (0.33 1.8 0110011 1.0e+09 1.0e-10
1.0e-10)

Pseudo-Random Bit Sequence Current Source

General Form:

IXYZ N+ N- PRBS(I1 I2 BR TR TF)

This form indicates a bit sequence of zeros and ones that is randomly generated. The parameters are listed below

Parameter	Unit
V1 (initial value,)	Amps
V2 (Final value)	Amps
XXXXYYY	Bit Sequence
BR (bit rate)	Bits/seconds
TR (rise time)	seconds
TF (fall time)	second

Example:

IP 1 0 PRBS (0.0 70 5.0e+08 1.0e-09 1.0e-09)

User Defined current Source

General Form:

```
IXYZ N+ N- USERDEF Filename
```

This form indicates a piecewise linear description of the voltage source can be read from a file. The points t_i and $i(t_i)$ are read directly from the file `Filename` in the format:

```
t1    i(t1)
t2    i(t2)
t3    i(t3)
```

The units are seconds and amperes. During a run, if the time specified is larger than the last time point, then the current value corresponding to the last time point is returned.

Example:

```
IP 1 0 USERDEF current_listing.txt
```

7. Linear Voltage-Controlled Current Sources – G-Element



General form:

```
GXYZ N+ N- NC+ NC- VALUE
```

Example:

```
G1 2 0 5 0 0.1MMHO
```

$N+$ and $N-$ are the positive and negative nodes, respectively. Current flow is from the positive node, through the source, to the negative node. $NC+$ and $NC-$ are the positive and negative controlling nodes, respectively. `VALUE` is the transconductance (in mhos).

8. Linear Voltage-Controlled Voltage Sources – E-Element



General form:

```
EXYZ N+ N- NC+ NC- VALUE
```

Example:

```
E1 2 3 14 1 2.0
```

$N+$ is the positive node, and $N-$ is the negative node. $NC+$ and $NC-$ are the positive and negative controlling nodes, respectively. `VALUE` is the voltage gain.

9. Linear Current-Controlled Current Sources – F-Element



General form:

FXYZ N+ N- VNAME VALUE

N+ and N- are the positive and negative nodes, respectively. Current flow is from the positive node, through the source, to the negative node. VNAME is the name of a voltage source through which the controlling current flows. The direction of positive controlling current flow is from the positive node, through the source, to the negative node of VNAME. VALUE is the current gain

Example:

F1 13 5 VSNS 5

10. Linear Current-Controlled Voltage Sources – H-Element



General form:


HXYZ N+ N- VNAME VALUE

N+ and N- are the positive and negative nodes, respectively. VNAME is the name of a voltage source through which the controlling current flows. The direction of positive controlling current flow is from the positive node, through the source, to the negative node of VNAME. VALUE is the transresistance (in ohms).

Examples:

HX 5 17 VZ 0.5K

11. Lossless Transmission Lines – T-Element



General Form


TXYZ N1 N1REF N2 N2REF Z0=VALUE TD=VALUE

N1 and N1REF are the signal and reference nodes at the near end (port 1) of the line; N2 and N2REF are the signal and reference nodes at the far end (port 2) of the line. Z0 is the characteristic impedance. TD is the transmission delay.

Example:

T1 1 0 2 0 Z0=50 TD=10NS

12. Lossy Transmission Lines – O-Element



General form:

OXYZ N1 N1REF N2 N2REF RLGCFILENAME LEN=LENGTH

N1 and N1REF are the signal and reference nodes at the near end (port 1) of the line; N2 and N2REF are the signal and reference nodes at the far end (port 2) of the line RLGCFILENAME is the name of the file containing the R, L, G and c parameters as a function of frequency.

Examples:

O23 1 0 2 0 LOSSYMOD
OCONNECT 10 5 20 5 INTERCONNECT

Format of RLGC file

The RLGC file containing the data for the transmission line is arranged as follows

```

Number of frequency points (M)
Frequency-factor R_factor L_factor G_factor C_factor
f1 R(f1)
f2 R(f2)
:
fk R(fk)
:
fM R(fM)
f1 L(f1)
f2 L(f2)
:
fk L(fk)
:
fM L(fM)
f1 G(f1)
f2 G(f2)
:
fk G(fk)
:
fM G(fM)
f1 C(f1)
f2 C(f2)
:
fk C(fk)
:


```

FM C(fM)

Actual values are obtained by multiplying the parameters by their respective factors. The parameters Frequency-factor R_factor L_factor G_factor C_factor refer to the scale of the units in which the parameters are read.

13. Coupled Lossless Multiconductor Lines - U element

In LIM, the lossless multiconductor line is represented by the U-element. This consists a group of M coupled transmission lines that have constant inductance and capacitance (self and mutual) parameters that do not change with frequency.



General form:

```

UXYZ N=M NN1 NN2 ... NNM NNREF NF1 NF2 .. NFM NFREF
LCFILE=filename L=length.

```

N=M is the number of coupled lines. NN1, NN2,.. are the signal nodes at the near end of the multiconductor system. NNREF is the reference node at the near end of the multiconductor system. NF1, NF2,. are the signal nodes at at the far end of the multiconductor system. NFREF is the reference node at the far end of the multiconductor system. FILENAME is the name of the file containing the inductance and capacitance matrices [L] and [C]. These are M by M matrices that contains the Maxwellian elements.

Examples:

```

U1 N=6 2 3 4 5 6 7 0 8 9 10 11 12 13 0 LCFILE=UELL L=14

```

Format of LC file

The LC file containing the data for the multiconductor system contains the Maxwellian inductance and capacitance lower-left triangular matrix elements.

```

+ Number of lines (N)
+ INFINITY
+ L11
+ L21
+ L22


```

```
+ ...
+...
+ LNN
+ INFINITY
+ C11
+ C21
+ C22
+ ...
+...
+ CNN
```

The line “+INFINITY” is to indicate that the inductance and capacitance values correspond to those for the infinite frequency limit.

14. Coupled Lossy Multiconductor Lines - W element

The W element is used to represent coupled transmission lines with frequency-dependent parameters. It can be used to simulate both lossy and lossless lines.



General form:

```
WXYZ N=Numlines NN1 NN2 ... NNM NNREF NF1 NF2 .. NFM NFREF
RLGCFILE=Filename L= LineLength.
```

Numlines is the number of lines, N_{1a} , N_{1b} , are the nodes tied to the near end and N_{2a} , N_{2b} , ..., N_{REF2} are the nodes tied to the far end. RLGCfile is the name of the file that contains the R, L, G, and C data as a function of frequency. LineLength is the length of the uniform coupled transmission lines with units using the MKSA system.

Examples:

```
W1 N=3 2 3 4 0 5 6 7 0 RLGCfile=WEL L=2.95
```

Format of RLGC file

The RLGC file containing the data for the multiconductor system is in the following format

```
+ Number of lines (N)
+ Number of frequency points(M)
+ frequency_1 ( $f_1$ )
```

```

+ R11(f1)
+ R12(f1)
+ ...
+ RNN(f1)
+ frequency_2(f2)
+...
+ RNN(f2)
+...
+ frequency_M(fM)
+...
+ RNN(fM)
+ frequency_1
+ L11(f1)
+ L12(f1)
+ ...
+ LNN(f1)
+ frequency_2(f2)
+...
+ LNN(f2)
+...
+ frequency_M(fM)
+...
+ LNN(fM)
+ frequency_1(f1)
+ G11(f1)
+ G12(f1)
+ ...
+ GNN(f1)
+ frequency_2(f2)
+...
+ GNN(f2)
+...
+ frequency_M(fM)
+...
+ GNN(fM)
+ frequency_1(f1)
+ C11(f1)
+ C12(f1)
+ ...
+ CNN(f1)
+ frequency_2(f2)
+...
+ CNN(f2)
+...
+ frequency_M(fM)
+...
+ CNN(fM)

```

R_{ij} , L_{ij} , G_{ij} and C_{ij} are the resistance, inductance, conductance and capacitance matrix elements per unit length respectively.

15. N element

The macromodel element is used to represent circuit blocks that are described by their frequency-dependent network parameters. This version will only process S parameters.

General form:

```
NXYZ NPORTS FILETYPE FILENAME N1 N2 N3 N4 ... NP
```

NPORTS is the number of ports

FILETYPE is a string that indicates the file type and can take one of two values: SPAR or POLERES. When the file type is SPAR, FILENAME is the name of the S-parameter file. N1 , N2 , . . . , NP is the list of nodes tied to the ports of the macromodel. When the file type is POLERES, FILENAME is the name of a data file containing the poles and residues of the macromodel. FILETYPE can be omitted in which case FILENAME will be assumed to be a S-parameter file.

Examples:

```
NMAC 2 SDATA/LOSSYTL.S2P 2 3
```

Format of S-Parameter file

The S parameter file containing the data for the macromodel and should be arranged in Touchstone format

The first line in the Touchstone file reads

```
!XYZ freq_unit format ports
```

XYZ is a string indicating a name or the file.

freq_unit is the unit for the frequency values.

format is a mnemonic for the format for the scattering parameter values. Valid formats are:

RI: real and imaginary

MA: magnitude(linear) and phase(degrees)

DB: magnitude (dB) and phase(degrees)

ports is the number of ports for the macromodel

All subsequent lines in the file that start with a "!" are treated as comment lines. The number of such lines is arbitrary. We then have a line in the format:

```
# freq_unit param format Ref Value
```

where freq_unit is the unit for the frequency values.

param is a letter indicating the type of parameter. In this version of LIM only "S" is allowed

format specifies the format in which the scattering parameters are listed according to the definitions given above.

Ref is a letter indicating the type of the reference

Value is the value of the reference system

After the headers, the data is listed on a line by line format where each line is associated with frequency:

```
frequency  param_11(a) param_11(b) param_12(a) param_12(b)  param_ij(a)
param_ij(b)
```

In this form, the param_ij refer to the network parameters associated with ports *i* and *j* of the macromodel. (a)/(b) refer to the respective component depending on the data format (eg real/imaginary or magnitude/phase) as defined on the first line.

Examples:

```
!_Coupled_Line Hz RI 4
!_Created_Thu_Feb_19_13:46:21_1998
#_hz_S_ri_R_50
!_4_port_Network_SP1.SP
!_Coupled_Line
1e+06 1.03389e-06 0.000629999 0.999999 -0.00112879 5.01921e-07 0.000327027 -4.43923e-07 -0.000210752
0.999999 -0.00112879 1.03389e-06 0.000629999 -4.43923e-07 -0.000210752 5.01921e-07 0.000327027
5.01921e-07 0.000327027 -4.43923e-07 -0.000210752 1.03389e-06 0.000629999 0.999999 -0.00112879
-4.43923e-07 -0.000210752 5.01921e-07 0.000327027 0.999999 -0.00112879 1.03389e-06 0.000629999

1.01e+08 0.00787306 0.0628062 0.990788 -0.113182 0.00504536 0.0323904 -0.00445385 -0.020664
0.990788 -0.113182 0.00787306 0.0628062 -0.00445385 -0.020664 0.00504536 0.0323904
0.00504536 0.0323904 -0.00445385 -0.020664 0.00787306 0.0628062 0.990788 -0.113182
-0.00445385 -0.020664 0.00504536 0.0323904 0.990788 -0.113182 0.00787306 0.0628062
```

Format of pole/residue file

The pole/residue file containing the data for the macromodel is arranged in the following format

The first line in the file reads

N port filename

In which N is the number of ports

The second line in the file reads

M poles

In which M is the total number of poles.

The third line in the file reads:

NC complex pairs

In which NC is the number of pairs of complex poles.

The fourth line in the file is in the form:

NR real poles

In which NR is the number of real poles.

The fifth line reads

F frequency factor

In which F is the scaling factor by which the poles will need to be scaled in order to be read in GHz

The sixth line reads

FM maximum frequency

In which FM is the maximum frequency in units scaled by F

The seventh line reads

Reference impedance matrix.

The next N2 lines will be the entries of the impedance matrix in ohms

The next M lines will have the poles listed as:

i real imag

In which i is the index of the pole, real and imag are the real and imaginary parts of the pole respectively.

The following line will read

residue i j

Indicating that the M complex residues of S_{ij} will be listed

The next 2XM lines list the residues of S_{ij} as

i real imag

In which i is the index of the residue, real and imag are the real and imaginary parts of the residue respectively.


Examples:

```
2 port EX1.S2P_poles_and_residues
185 poles
92 complex pairs
1 real poles
1.00000e+09 frequency factor
```

```
4.00000e+01 maximum frequency
reference impedance matrix
5.00000e+01
0.00000e+00
0.00000e+00
5.00000e+01
1 -1.9739e+09 2.5010e+11
2 -1.9739e+09 -2.5010e+11
3 -1.0349e+09 2.4633e+11
:
184 -2.2690e+09 -2.4715e+09
185 -1.70004e+09
residues 1 1
1 7.4953e+08 -6.0715e+08
2 7.4953e+08 6.0715e+08
3 2.7926e+07 1.0057e+08
:
185 1.85419e+08
constant term -2.66615e-02
residues 1 2
1 1.2977e+08 1.0013e+08
2 1.2977e+08 -1.0013e+08
:
:
constant term 5.61591e-02
```

16. IBIS Models - P element

The P element is used to represent ibis behavioral models.



General form:

```
PXYZ IBISFILE.IBS T E IO PU PD PC GC
```

IBISFILE.IBS is the name of the ibis file. This file follows the prescribed ibis format. The nodes are defined as follows:

T: trigger - the buffer state is controlled by the voltage between this terminal and ground.

E: enable - enabling and disabling the buffer is controlled by the voltage between this terminal.

IO: input/output - when io is disabled, it is the input terminal when IO is enabled, it is the output terminal

PU: pull-up - power supply terminal for the pullup (PU) device.

PD: pull-down - power supply terminal for the pulldown (PD) device.

PC: power clamp - power supply terminal for the power clamp (PC)


GC: ground clamp - power supply terminal for the ground clamp (GC)

Refer to the ibis forum (<http://ibis.org>) for information related to the ibis file format. The LIM simulator contains an IBIS parser that will read the specified IBIS file and extract all the relevant parameters needed for the analysis

Examples:

```
PIBIS K4T1G164QE.IBS 17 18 16 19 20 21 22
```

17. Diodes – D-Element



General form:

DXYZ N+ N- MNAME

N+ and N- are the positive and negative nodes, respectively. MNAME is the model name. The general equation for the diode model is given by

$$I_D = I_S \left(e^{\frac{qV_D}{kT}} - 1 \right) - I_Z e^{A(V_b - B'V_D)} + \frac{V_D}{R_C}$$

DIODE MODEL PARAMETERS


	Param	Description	Unit	Default
1	IS	saturation current (I	A	10 ⁻¹⁴
2	RS	ohmic resistance	W	0
3	N	emission coefficient	-	1
4	TT	transit-time	sec	0
5	CJO	zero-bias junction capacitance	F	0
6	VJ	junction potential	V	1
7	M	grading coefficient	-	0.5
8	EG	activation energy	eV	1.11
9	XTI	saturation-current temp. exp	-	3
10	KF	flicker noise coefficient	-	0
11	AF	flicker noise exponent	-	1
12	FC	coefficient for forward-bias	-	0.5
13	BV	reverse breakdown voltage	V	inf
14	IBV	current at breakdown voltage	A	10 ⁻³
15	TNOM	parameter measurement temperature	C	27
16	RC	Zener resistance (R_C)		inf
17	A	Exponential factor (A)		1
18	BP	B attenuation constant (B')		1.0
19	VB	Zener voltage(V_b)	V	0
20	IZ	Zener saturation current (I_z)	A	0

MNAME is the name of the diode model which must be found in a .MODEL line. Such a line will have the form:

```
.MODEL MNAME D IS=Value N=1 XTI= Value RC=value A=1.0 BP=1.15
+ VB=-18.0 IZ=1.0
```

Examples:

```
DBRIDGE 2 10 DIODE1
DCLMP 3 7 DMOD 3.0 IC=0.2
```

18. Bipolar Junction Transistors (BJTs) – Q-Element**General form:**

```
QXYZ NC NB NE MNAME
```

NC, NB, and NE are the collector, base, and emitter nodes, respectively. NS is the (optional) substrate node. If unspecified, ground is used. MNAME is the model name. Details on the format of that file will be given later.


Examples:

```
Q23 10 24 13 QMOD IC=0.6, 5.0
Q50A 11 26 4 20 MOD1
```

BJT EBERS-MOLL MODEL PARAMETERS

	Param	Description	Unit	Default
1	BF	forward beta	-	100
2	BR	reverse beta	-	1
3	IS	saturation current	A	10^{-16}
4	VAF	forward Early voltage	V	infinite
5	VAR	reverse Early voltage	V	infinite
6	RB	zero bias base resistance	ohms	0
7	RC	collector resistance	ohms	0
8	RE	emitter resistance	ohms	0
9	CJE	base-emitter zero-bias depletion capacitance	F	2×10^{-12}
10	TR	reverse transit time	sec	0
11	TF	ideal forward transit time	sec	0
12	CJC	base-collector zero-bias depletion capacitance	F	0

19. MOSFETs



General form:

```
MXYZ ND NG NS NB MNAME L=VAL W=VAL
```

ND, NG, NS, and NB are the drain, gate, source, and bulk (substrate) nodes, respectively. MNAME is the model name. L and W are the channel length and width, in meters.

Example:

```
M1 24 2 0 20 TYPE1
M31 2 17 6 10 MODM L=5U W=2U
M1 2 9 3 0 MOD1 L=10U W=5U
```

Models for MOSFETS

The .MODEL statement for a MOSFET reads

```
.MODEL MODELNAME POLARITY LEVEL=VALUE P1=VAL P2=VAL ...PN=VAL
```

.MODEL where MODELNAME is the name of the model specified in the element line, POLARITY is either “NMOS” for N-channel devices or “PMOS” for P-channel devices. LEVEL specifies the level number for the model, next the PJ=VAL specify the values of some of the parameters associated with the model level.

MOSFET LEVEL 1 MODEL PARAMETERS

	Parameter	Description	Unit	Default
1	UO	Surface mobility	cm ² /V	600
2	UCRIT	Critical field for mobility degradation	V/cm	
3	UEXP	critical field exponent in mobility degradation -		
4	TOX	oxide thickness	meter	10 ⁻⁷
5	XJ	metallurgical junction depth	meter	0
6	CGSO	gate-source overlap capacitance per meter channel width	F/m	0
7	CGDO	gate-drain overlap capacitance per meter channel width	F/m	0
8	CBD	zero-bias B-D junction capacitance	F	0
9	CBS	zero-bias B-S junction capacitance	F	0
10	LD	lateral diffusion	meter	0
11	NSS	surface state density	1/cm ²	0

12	KP	transconductance parameter	A/V^2	2.0e-5
13	PHI	surface potential	V	0.6
14	VTO	zero-bias threshold voltage (VTO)	V	0
15	DELTA	width effect on threshold voltage -	0	
16	GAMMA	bulk threshold parameter	$V^{1/2}$	0
17	LAMBDA	channel-length modulation	$1/V$	0
18	ETA	static feedback -	0.0	
19	KAPPA	saturation field factor	-	0.2
20	THETA	mobility modulation	$1/V$	0
21	MJ	bulk junction bottom grading coeff.	-	0.5
22	MJSW	bulk junction sidewall grading coeff.	-	0.5
23	CJ	zero-bias bulk junction bottom cap.per sq-meter of junction area	F/m^2	0
24	CJSW	zero-bias bulk junction sidewall cap. per meter of junction perimeter	F/m	0
25	PB	bulk junction potential	V	0.8

Level 1 is the most basic model employed by LIM. It is based on the Shichman-Hodges model [6]. It is used mainly for long-channel devices. The charge-storage dynamic behavior for this levels employs the Meyer model [7], which is not a charge-conserved model.


LIM also supports more advanced levels based on the BSIM (Berkeley Short-Channel IGFET Model) project. These models employ empirical parameters and polynomial equations to handle various physical effects. This generally leads to improved circuit simulation behavior compared to earlier models. It becomes less accurate for although its accuracy degrades in submicron FETs. Currently available in LIM are BSIM3 (Level 37) and BSIM4 (Level 54) models. An extensive list of the BSIM3 and BSIM4 parameters and their default values can be found in the literature. BSIM models belong to the category of the third generation of MOSFET models that fulfilled two main purposes. The first is the need for models suitable for analog design. The second is to reduce the number of model parameters from that found in the second-generation models, and return more physical meaning to those parameters.

There are some limitations in the models. At this stage accuracy and convergence are not guaranteed for any combination of device parameters. Furthermore, the polynomial equations used in these models can behave poorly, causing negative electrical parameters and convergence problems. The improvement of the algorithms and models is an ongoing effort.

LIM also supports (in beta mode) the FinFET model based on the BSIM-CMG. Due to their excellent low power and scaling characteristics, FinFET devices have been adopted in all the MOS technology nodes below 20 nanometers. For this technology, the Compact Model Coalition (CMC) has selected BSIM-CMG as the first and only industry-standard compact model for advanced circuit design. BSIM-CMG compact model includes several physical effects: device geometry effects, charge quantization, gate oxide tunneling, gate capacitance degradation, short-channel effects. The table below shows the different models supported by LIM for various MOS technologies and their associated level number.

MOS MODELS AND LEVELS SUPPORTED BY LIM

Technology	Model	Level
>1000 nm	Shichman-Hodges	1
180 nm	BSIM3	37
130 nm	BSIM3	37
90 nm	BSIM4	54
65 nm	BSIM4	54
45 nm	BSIM4	54
32 nm	BSIM4	54
22 nm	BSIM4	54
FinFet (< 10 nm)	BSIM-CMG	57

20. Nonlinear Voltage-Controlled (Current or Voltage) Source – B-Element**General form:**

BXYZ N+ N- $VA=f(VB, VC, VD, \dots)$

BGHF N+ N- $IA=f(VB, VC, VD, \dots)$

Example:

BEL1 4 0 $V=V(3)-0.3*V(2)^3$

N+ is the positive node, and N- is the negative node. The formula on the right-hand side of the equal sign expresses the value of the voltage or current as a nonlinear function of one or several controlling node voltages. The value resulting from the expression for the V and I variables are the voltages and currents across or through the nonlinear element, respectively. If I is used then the element is a current source, and if V is used the device is a voltage source. One and only one of these variables must be used.

For the expression, the following standard operators are defined: +, -, *, /, ^.

In addition, a list of transcendental functions of real variables are also available for use. These functions are: *abs*, *asinh*, *cosh*, *sin*, *acos*, *atan*, *exp*, *sinh*, *acosh*, *atanh*, *ln*, *sqrt*, *asin*, *cos*, *log*, *tan*.


Example 1 - NMOS Inverter

```

*** NDINV * 4
MJOS 3 1 4 0 N342 L=1.2U W=5U
RS 4 0 2000
RD 2 3 10000
RG 1 5 50.0
VDD 2 0 5.0
CDD 2 0 1.E-18
VP 5 0 PULSE (0 5 0.20000e-09 0.10000e-09 0.10000e-09 2.00000e-09 6.00000e-09)
.TRAN 3.12500e-12 4.00e-09
.PRINT TRAN V(1) V(2) V(3) V(4)
.PLOT TRAN V(1) V(2) V(3) V(4)


*** 0.5um CMOS technology
*SPICE LEVEL3 PARAMETERS
.MODEL N342 NMOS LEVEL=3 PHI=0.7 TOX=9.5E-09 XJ=0.2U TPG=1
+ VTO=0.7 DELTA=8.8E-01 LD=5E-08 KP=1.56E-04
+ UO=420 THETA=2.3E-01 RSH=2.0E+00 GAMMA=0.62
+ NSUB=1.40E+17 NFS=7.20E+11 VMAX=1.8E+05 ETA=2.125E-02
+ KAPPA=1E-01 CGDO=3.0E-10 CGSO=3.0E-10
+ CGB0=4.5E-10 CJ=5.50E-04 MJ=0.6 CJSW=3E-10
+ MJSW=0.35 PB=1.1
.END

```



Example 2 - Diode Clipper

```
*MAIN CIRCUIT
VIN 1 0 SIN (0 12.0 41.66E+06 0.0 0.0 )
RG1 1 2 8000.0
D1 2 3 DIODE
D2 4 2 DIODE
VDC1 3 0 3.0
VDC2 0 4 1.0
RG2 3 0 500.0
.TRAN 2.3N 45.60000N
.PLOT TRAN V(2) V(3)
.END
```



Example 3 - 2-Input NAND Gate


* cmos nand

```

M1 5 1 0 0 N342 L=5U W=5U
M2 3 2 5 5 N342 L=5U W=5U
M3 3 1 4 4 P342 L=5U W=5U
M4 3 2 4 4 P342 L=5U W=5U
C1 5 0 1e-14
C2 3 0 1e-14
VA1 2 0 PWL (0 0 5e-08 0 6e-08 6 11e-08 6 12e-08 0 15e-08 0)
VB1 1 0 PWL (0 0 2e-08 0 3e-08 6 5e-08 6 6e-08 0 8e-08 0 9e-08 6 11e-08 6 12e-08 0 15e-08
0)
VDD 4 0 6
.TRAN 1e-10 1.4e-07
.PRINT TRAN V(1) V(2) V(3) V(4)
.PLOT TRAN V(1) V(2) V(3) V(4)
*** 0.5um CMOS technology
*SPICE LEVEL3 PARAMETERS
.MODEL N342 NMOS LEVEL=3 PHI=0.7 TOX=9.5E-09 XJ=0.2U TPG=1
+ VTO=0.7 DELTA=8.8E-01 LD=5E-08 KP=1.56E-04
+ UO=420 THETA=2.3E-01 RSH=2.0E+00 GAMMA=0.62
+ NSUB=1.40E+17 NFS=7.20E+11 VMAX=1.8E+05 ETA=2.125E-02
+ KAPPA=1E-01 CGDO=3.0E-10 CGSO=3.0E-10
+ CGBO=4.5E-10 CJ=5.50E-04 MJ=0.6 CJSW=3E-10
+ MJSW=0.35 PB=1.1

*** 0.5um CMOS technology
*SPICE LEVEL3 PARAMETERS
.MODEL P342 PMOS LEVEL=3 PHI=0.7 TOX=9.5E-09 XJ=0.2U TPG=-1
+ VTO=-0.95 DELTA=2.5E-01 LD=7E-08 KP=4.8E-05
+ UO=130 THETA=2.0E-01 RSH=2.5E+00 GAMMA=0.52
+ NSUB=1.0E+17 NFS=6.50E+11 VMAX=3.0E+05 ETA=2.5E-02
+ KAPPA=8.0E+00 CGDO=3.5E-10 CGSO=3.5E-10
+ CGBO=4.5E-10 CJ=9.50E-04 MJ=0.5 CJSW=2E-10
+ MJSW=0.25 PB=1
.END

```



Example 4 - CMOS Inverter


```

*** CMOSINV * 4
MJOS 4 2 0 0 N342 L=1.2U W=5U
MSCH 4 2 3 3 P342 L=1.2U W=5U
RG 1 2 50
RD 4 0 10000
VDD 3 0 5
CDD 3 0 1.E-06
VP 1 0 PULSE (0 5 0.10000e-09 0.10000e-09 0.10000e-09 2.00000e-09 6.00000e-09)
.TRAN 3.12500e-11 4.00e-09
.PRINT TRAN V(1) V(2) V(3)

**** LEVEL 1 NMOS ****
.MODEL N342 NMOS
+ LEVEL=1 TPG=1
+ KP=3.6E-05
+ LAMBDA=0.00033 VT0=0.69486 GAMMA=0.60309 PHI=1
+ TOX=1.980000E-08 XJ=0.2U LD=0.1U NSUB=4.9999999E+16
+ NSS=0.000000E+00
+ CJ=4.091E-4 MJ=0.307 PB=1.0
+ CJSW=3.078E-10 MJSW=1.0E-2
+ CGSO=3.93E-10 CGDO=3.93E-10

**** LEVEL 1 PMOS ****
.MODEL P342 PMOS
+ LEVEL=1 TPG=-1
+ KP=7.69968E-06
+ LAMBDA=0.00033 VT0=-0.69486 GAMMA=0.89213 PHI=1
+ TOX=1.980000E-08 XJ=0.2U LD=0.1U NSUB=4.9999999E+16
+ NSS=0.000000E+00
+ CJ=6.852E-4 MJ=0.429 PB=1.0
+ CJSW=5.217E-10 MJSW=0.351
+ CGSO=7.29E-10 CGDO=7.29E-10
.END

```



SPICE vs LIM Elements

Element	SPICE	LIM	Comment
A	NOT USED	Macromodel-convolution	
B	Nonlinear dependent source	Nonlinear dependent source	
C	Capacitor	Capacitor	
D	Diode	Diode	
E	Voltage-dependent voltage src	Voltage-dependent voltage src	
F	Current-dependent current src	Current-dependent current src	
G	Voltage-dependent current src	Voltage-dependent current src	
H	Current-dependent voltage src	Current-dependent voltage src	
I	Independent current source	Independent current source	
J	JFET	NOT YET JFET	
K	Mutual inductance	Mutual inductance	
L	Inductance	Inductance	
M	MOSFETS	MOSFETS	
N	NOT USED	Macromodel-MOR	
O	Lossy TL	Lossy Transmission Line	
P	NOT USED	IBIS	
Q	BJT	BJT	
R	Resistor	Resistor	
S	Switches	NOT YET Switches	
T	Lossless Transmission Line	Lossless Transmission Line	
U	NOT USED	Multi-Coupled TL	
V	Independent Voltage Source	Independent Voltage Source	
W	W-Element (HSPICE)	W-Element (F-dep MTL)	
X	Subcircuit	Subcircuit	
Y	NOT USED	Nonlinear Element (IV Table)	
Z	MESFETS	NOT YET - MESFETS	